

TILASTOLLISET MENETELMÄT TÄHTITIEESSÄ

IDL-harjoitus 5, Heikki Salo 17.4.2008

ESIMERKKIVASTAUKSET

1. KS-testi: vertaa otosta ja oletettua teoreettista jakaumaa


Tutkitaan noudattaako Saturnuksen renkaiden simulaatiossa saatu partikkelien vertikaalinen jakauma Gaussista jakaumaa. Seuraaviin IDL-save tiedostoihin on tallennettu partikkelien paikat kadessa eri simulaatiossa eräällä ajanhetkellä (edustaa tasapainotilaa):

tb_iimodel2_a100tau010.pos - harva systeemi (optinen paksuus 0.1; C-ring)
tb_iimodel1_a100tau400.pos - tihea systeemi (optinen paksuus 4.0; B-ring)


a) Lue data IDL:n restore-komennolla. Esim. partikkelien x,y,z-koordinaatit muuttujissa 'xx','yy','zz'.

```
IDL> restore,'tb_iimodel2_a100tau010.pos',/v
% RESTORE: Portable (XDR) SAVE/RESTORE file.
% RESTORE: Save file written by heikki@halla.oulu.fi, Sun Apr 27 19:48:23 2003.
...
% RESTORE: Restored variable: XX.
% RESTORE: Restored variable: YY.
% RESTORE: Restored variable: ZZ.
...
```

HARVA RENGAS (tb_iimodel2_a100tau010.pos)



TIHEA RENGAS (tb_iimodel1_a100tau400.pos)



b) Testaa Kolmogorov-Smirnov-testin avulla noudattaako harvan systeemin z:n jakauma Gaussista jakaumaa, jonka keskiarvo = 0 ja hajonta sama kuin z:n otoshajonta. (IDL: KSTONE-funktio).


c) Toista testi tiheälle systeemille.

- Nollahypoteesi H_0 : otos noudattaa Gaussista jakaumaa
- Testisuure: D = otoksen ja teoreettisen mallin kumulatiivisten tiheysfunktioiden maksierotus

Käytetään IDL:n ksone-proceduuria, joka palauttaa testisuureen D arvon, sekä todennäköisyyden että hypoteesin H_0 ollessa voimassa saadaan havaitun suuruinen tai vielä suurempi D :n arvo.

Hylätään H_0 , jos saatu todennäköisyys on pienempi kuin ennalta määrätty merkitsevyystaso (esim. 0.01). Muussa tapauksessa H_0 jää voimaan.

Huom: KSTONE palauttaa kaksisuuntaisen testin todennäköisyyden (eli ei ota kantaa mihin suuntaan otos ja malli mahdollisesti poikkeavat)



Johtapäätökset:

- harvan systeemin vertikaalinen tiheysjakauma Gaussinen ($p = 0.94 \gg 0.01$)
- tiheän systeemin ei ole Gaussinen ($p = 0.0076 < 0.01$)

```

;-----
;Teoreettinen malli: Gaussinen jakauma, mu=0, sig=sigma
;KSONE:n kutuma aliohjelma
;-----
function ks_model,x, _extra=extra
    common ks_model_com,sigma
    res=gauss_pdf(x/sigma)
    return,res
end
;-----
;PAAOHJELMA
;Testataan hypoteesia H0: partikkelien vertikaalinen jakauma
;noudattaa Gaussista jakaumaa
;Partikkelit Saturnuksen renkaiden simulaatiosta
;-----
    common ks_model_com,sigma
    program='idlharj5_ksone_zz' & ps=0.1
    psdirect,program,ps,/color
    if(!d.name eq 'PS') then !p.charsize=0.7

;-----
;1000 particles
;HARVA SYSTEEMI
    restore,'tb_iimodel2_a100tau010.pos'
    data=zz
    sigma=sqrt(mean(data^2))

    !p.multi=[0,2,2] & nwin & ff='(f6.4)'
    !x.range=[-20,20]
    !p.title='1000 z arvoa: harva systeemi'
    histo_f,data,-30,30,1,/plot,/gauss,psym=10
    xyouts,5,0.08,'sigma='+string(sigma,ff)

    !p.title='ACCEPT GAUSSIAN'
    !x.range=[-20,20]*.5
    ksone,data,'ks_model',d,prob,/plot,xtitle='z/radius',ytitle='CFD'
    xyouts,-8,0.8,'D='+string(d,ff)
    xyouts,-8,0.7,'Prob='+string(prob,ff)

;-----
;1000 particles
;TIHEA SYSTEEMI
    restore,'tb_iimodel1_a100tau400.pos'
    data=zz
    sigma=sqrt(mean(data^2))

    !x.range=[-20,20]
    !p.title='1000 z arvoa: tihea systeemi'
    histo_f,data,-30,30,1,/plot,/gauss,psym=10
    xyouts,5,.08,'sigma='+string(sigma,ff)

    !p.title='REJECT GAUSSIAN'
    !x.range=[-20,20]*.5
    ksone,data,'ks_model',d,prob,/plot,xtitle='z/radius',ytitle='CFD'
    xyouts,-8,0.8,'D='+string(d,ff)
    xyouts,-8,0.7,'Prob='+string(prob,ff)

    !p.multi=0 & defplot
    psdirect,program,ps,/stop
    !p.charsize=0.7
end

```

2. KS-testi: vertaa kahta otosta (WJ Exercise 5.1)

Tiedosto “5point2.dat” sisältää radioaallonpituudella tehdyt vuontiheys-mittaukset 386 galaksille (“GAL”), sekä 290 vertailumittausta, joissa teleskooppi on osoittanut tyhjälle taivaan alueelle (“SKY”).

a) Lue data

Tiedosto näyttää seuraavalta: (alussa ei ole tyhjää riviä)

Galaxy positions, 386 measurements, format (10f8.3)


```
6.343 0.359 12.991 7.706 0.467 3.056 4.675 1.813 -2.552 -3.149
-0.972 -3.000 3.989 5.404 -3.515 -1.727 10.154 5.013 -1.093 -6.886
....
-10.255 13.393 13.220 -11.376 13.105 12.565
```

Random positions, 290 measurements, format (10f8.3)

```
-1.200 -4.804 1.077 6.444 -3.748 1.316 2.683 -5.263 6.755 1.538
...
3.945 3.830 4.310 9.604 11.760 -9.962 -10.396 11.702 11.765 11.789
```

Luetaan IDL:ssä esimerkiksi seuraavalla tavalla:

```
;-----
;read WJ data in 5point2.dat
;-----
file='5point2.dat'
close,1
openr,1,file
line=''
readf,1,line
readf,1,line
gal=fltarr(386)
readf,1,gal
readf,1,line
readf,1,line
readf,1,line
sky=fltarr(290)
readf,1,sky
close,1
;Nopea tarkistus:
histo_f,gal,-20,20,1,/nos,/plot
histo_f,sky,-20,20,1,/nos,/oplot,psym=10
```



b) Tutki hypoteesia, jonka mukaan galakseista tuleva vuontiheys on suurempi kuin vertailumittauksissa, käyttäen Kolmogorov-Smirnov testiä. (Huom: halutaanko 1- vai 2-suuntainen todennäköisyys?)


- Nollahypoteesi H_0 : GAL ja SKY otokset noudattavat samaa jakaumaa
- Testisuure: D = otoksien kumulatiivisten tiheysfunktioiden maksierotus

Käytetään IDL:n `kstwo`-proseduuria, joka palauttaa testisuureen D arvon, sekä todennäköisyyden että hypoteesin H_0 ollessa voimassa saadaan havaitun suuruinen tai vielä suurempi D :n arvo.

Hylätään H_0 , jos saatu todennäköisyys on pienempi kuin ennalta määrätty merkitsevyystaso (esim. 0.01). Muussa tapauksessa H_0 jää voimaan.

Huom: `KSTWO` palauttaa kaksisuuntaisen testin todennäköisyyden. Nyt halutaan yksisuuntaisen todennäköisyys, sillä vaihtoehtoinen hypoteesi on että GAL otoksen vuontiheydet suurempia kuin SKY otoksen. Yksisuuntaisen testin todennäköisyyttä voidaan approksimoida jakamalla kaksisuuntaisen testin antama todennäköisyys tekijällä 2. Tarkempi arvio voidaan tehdä luennoilla esitetyllä tavalla.

Esimerkkiohjelman `idlharj5_kstwo_WJ.pro` tulostaa:



Johtopäätös:

GAL ja SKY otoksien välillä ei ole tilastollisesti merkittävää eroa ($p(D|H_0) = 0.22 \gg 0.01$)

```

program='idlharj5_kstwo_WJ' & ps=0.1

;-----
;read WJ data in 5point2.dat
;-----
    file='5point2.dat'
    close,1
    openr,1,file
    line=''
    readf,1,line
    readf,1,line
    gal=fltarr(386)
    readf,1,gal
    readf,1,line
    readf,1,line
    readf,1,line
    sky=fltarr(290)
    readf,1,sky
    close,1
;-----
    title='WJ EXERCISE 5.1'
    psdirect,program,ps,/col
    coll=1 & if(!d.name eq 'PS') then coll=0
;-----
;plot histograms
;histo_f,x,x1,x2,dx,xx,yy,gg
    histo_f,gal,-20,20,1,xt,yt_gal,/nos
    histo_f,sky,-20,20,1,xt,yt_sky,/nos
    nwin
    plot,xt,yt_gal,xtitle='flux density mJy',ytitle='Number',psym=10,$
        title=title,yr=[0,50],ys=1,thick=2
    oplot,xt+.02,yt_sky+.2,psym=10,lines=2,col=2,thick=3
    label_data,0.55,0.95,['GALAXY','SKY'],lines=[0,2],col=[coll,2]
;KSTWO
; Return the Kolmogorov-Smirnov statistic and associated 2-tailed probability
; that two arrays of data values are drawn from the same distribution
    kstwo,gal,sky,d,prob
    print,'-----'
    print,'kstwo:'
    print,'Maximum D',d
    print,'Probability in two tailed test',prob
;One-tailed test-statistic
    m=n_elements(gal)
    n=n_elements(sky)
    chi2=4*d^2*m*n/(m+n)
    print,'chi2-statistic', chi2
    print,'probability in one-tailed test:', 1-chisqr_pdf(chi2,2)

    xyouts,-18,38,'KS-test: (kstwo.pro)'
    xyouts,-18,36,'max D'+string(d,'(f6.3)')
    xyouts,-18,34,'1-tailed significance: (gal>sky)'
    xyouts,-16,32,string(1-chisqr_pdf(chi2,2),'(f6.3)')
    psdirect,program,ps,/stop
end

```

3. U-testi: vertaa kahta otosta (WJ Exercise 5.2)

Toista tehtävässä 2 tehty tarkastelu, käyttäen Wilcoxon-Mann-Whitney U-testiä.

- Luetaan data kuten edellä
- Sovelletaan U-testiä (IDL: `rs_test.pro`)

```
-----  
;Apply Wilcoxon-Mann-Whitney test  
;  
;   RS_TEST(X,Y) tests the hypothesis that two sample populations,  
;   {X[i], Y[i]}, have the same mean of distribution against the  
;   hypothesis that they differ. The result is a two-element vector  
;   containing the nearly-normal test statistic Z and the one-tailed  
;   probability of obtaining a value of Z or greater. This type of  
;   test is often referred to as the Wilcoxon Rank-Sum Test or Mann-  
;   Whitney U-Test.  
  
   res=rs_test(sky,gal)  
   Z=res(0)  
   prob=res(1)  
   print,'-----'  
   print,'rs_test:'  
   print,'Z',Z  
   print,'Probability in one-tailed test',prob  
  
-----  
rs_test:  
Z      0.341223  
Probability in one-tailed test      0.366468
```

Kuten edellä, H_0 (otoksien välillä ei eroa) jää voimaan ($p(D|H_0) = 0.366 \gg 0.01$)

4. χ^2 -testi: vertaa kahta otosta (WJ Exercise 5.2)

Toista tehtävässä 2 tehty tarkastelu, käyttäen kahden otoksen χ^2 -testiä.

Esimerkkiohjelmassa `idlharj5_CTI_WJ.pro` käytetään IDL:n `cti_test.pro` ohjelmaa, seuraten Wall-Jenkins tehtävää 5.2.

Ohessa on otoksien frekvenssitaulukot (katetty 1 mJy tabulointia), sekä kutakin taulukon valia vastaavat odotetut frekvenssit, sekä näistä muodostettu kontribuutio χ^2 testisuureeseen. Kokonais χ^2 saadaan summaamalla kaytetyt taulukkovälit.

x	GAL_OBS	GAL_EXP	CHI2	SKY_OBS	SKY_EXP	CHI2
-13.0000	1	0.571006	0.322301	0	0.428994	0.428994
-12.0000	2	1.14201	0.644602	0	0.857988	0.857988
-11.0000	5	2.85503	1.61151	0	2.14497	2.14497
-10.0000	7	5.71006	0.291406	3	4.28994	0.387872
-9.00000	2	4.56805	1.44370	6	3.43195	1.92161
-8.00000	11	7.42308	1.72359	2	5.57692	2.29416
-7.00000	11	9.70710	0.172203	6	7.29290	0.229208
-6.00000	7	10.2781	1.04552	11	7.72189	1.39163
-5.00000	13	14.2751	0.113904	12	10.7249	0.151611
-4.00000	11	16.5592	1.86630	18	12.4408	2.48411
-3.00000	25	25.1243	0.000614567	19	18.8757	0.000818009
-2.00000	39	35.4024	0.365596	23	26.5976	0.486621
-1.00000	31	29.1213	0.121200	20	21.8787	0.161322
0.00000	32	32.5473	0.00920436	25	24.4527	0.0122513
1.00000	37	36.5444	0.00568047	27	27.4556	0.00756097
2.00000	22	27.9793	1.27780	27	21.0207	1.70079
3.00000	31	30.8343	0.000890237	23	23.1657	0.00118494
4.00000	23	25.6953	0.282716	22	19.3047	0.376305
5.00000	21	18.2722	0.407228	11	13.7278	0.542035
6.00000	11	15.4172	1.26556	16	11.5828	1.68450
7.00000	6	9.13610	1.07651	10	6.86391	1.43287
8.00000	16	9.13610	5.15682	0	6.86391	6.86391
9.00000	4	2.28402	1.28920	0	1.71598	1.71598
10.0000	6	6.28106	0.0125771	5	4.71894	0.0167405
11.0000	4	2.28402	1.28920	0	1.71598	1.71598
12.0000	2	3.42604	0.593566	4	2.57396	0.790056
13.0000	6	3.42604	1.93381	0	2.57396	2.57396


χ^2 testin edellytykset ovat voimassa vain jos kaikissa soluissa odotettu frekvenssi ≥ 1 ja vähintään 80% odotetuista frekvensseistä ≥ 5 .

1) Otetaan kaikki solut joissa odotetut frekvenssit $\geq 1 \Rightarrow$ saadun χ^2 todennäköisyys $p = 0.00037$, eli otosten välillä näyttäisikin olevan merkittävä ero! (Edellä KS ja U-testit puolsivat voimakkaasti sitä että eroa ei ole).

2) Poistetaan solut $x \leq -11$ ja $x \geq 11$, joissa odotetut frekvenssit pieniä \Rightarrow saadun χ^2 todennäköisyys $p = 0.0033$, eli vieläkin näyttäisi olevan merkittävä ero!

3) Suurin osa χ^2 peräisin solusta $x = 8$. Ehkä tämä on jotenkin 'epätyypillinen' tai 'huono' solu? Poistetaan solut $x \leq -8$ ja $x \geq 8 \Rightarrow$ todennäköisyys $p = 0.28$. Eli nyt ero otosten välillä näyttää hävinneen, kuten KS ja U testeissä.

```
*kaytetaan kaikki valit joille GAL_EXP ge 1 and SKY_EXP ge 1:
  chi2=    55.9253  DF=      25
  prob=    0.000371635
*kaytetaan vali -10 -> 10
  chi2=    41.3275  DF=      20
  prob=    0.00338060
*kaytetaan vali -7 -> 7
  chi2=    16.4807  DF=      14
  prob=    0.284910
```



JOHTAPÄÄTÖS: testin soveltamisessa ollaan menty vaarallisen subjektiiviselle linjalle! Mitään johtopäätöstä ei voi tehdä!

```

program='idlharj5_CTI_WJ' & ps=-1
psdirect,program,ps,/col
!p.charsize=0.7

;-----
;read WJ data in 5point2.dat
file='5point2.dat'
close,1
openr,1,file
line=' '
readf,1,line
readf,1,line
gal=fltarr(386)
readf,1,gal
readf,1,line
readf,1,line
readf,1,line
sky=fltarr(290)
readf,1,sky
close,1
;-----
!p.multi=[0,1,3]
;lasketaan frekvenssitaulukot
histo_f,gal,-15.5,15.5,1,x_gal,y_gal,/noscale
histo_f,sky,-15.5,15.5,1,x_sky,y_sky,/noscale
;yhdistetaan yhteen taulukkoon ja kaytetaan
;cti-test ohjelmaa odotettujen frekvenssien laskemiseen
obsfreq=[transpose(y_gal),transpose(y_sky)]
res=cti_test(obsfreq*1.,df=df,exfreq=exfreq)
;tulostetaan kaikki binit, jotta nahdaan miten chi^2 muodostuu
;ja mitka soult pitaisi hylata liian pinene frekvenssin takia
print,'-----'
print,'          x          GAL_OBS    GAL_EXP    CHI2          SKY_OBS    SKY_EXP    CHI2'
print,'-----'
for i=0,n_elements(x_gal)-1 do begin
    if(y_gal(i) gt 0 or y_sky(i) gt 0) then begin
        chisq1=(obsfreq(0,i)-exfreq(0,i))^2/exfreq(0,i)
        chisq2=(obsfreq(1,i)-exfreq(1,i))^2/exfreq(1,i)
        print,x_gal(i),y_gal(i),exfreq(0,i),chisq1,y_sky(i),exfreq(1,i),chisq2
    endif
endfor
print,'-----'

for icase=1,3 do begin
    histo_f,gal,-15.5,15.5,1,x_gal,y_gal,/noscale,/plot,psym=10
    histo_f,sky,-15.5,15.5,1,x_sky,y_sky,/noscale,/oplot
    if(icase eq 1) then begin
        print,'*kaytetaan kaikki valit joille GAL_EXP ge 1 and SKY_EXP ge 1:'
        ind=where(exfreq(0,*) ge 1 and exfreq(1,*) gt 0)
        endif
    if(icase eq 2) then begin
        print,'*kaytetaan vali -10 -> 10'
        ind=where(abs(x_gal) le 10)
        endif
    if(icase eq 3) then begin
        print,'*kaytetaan vali -7 -> 7'
        ind=where(abs(x_gal) le 7)
        endif
    oplot,x_gal(ind),y_gal(ind),psym=10,thick=5,col=2
    oplot,x_sky(ind),y_sky(ind),thick=5,col=2
;yhdistetaan otosfrekvenssit yhteen taulukkoon
    obsfreq=[transpose(y_gal(ind)),transpose(y_sky(ind))]
    res=cti_test(obsfreq*1.,df=df,exfreq=exfreq)
    print,' chi2=',res(0),' DF=',df
    print,' prob=',res(1)
    xyouts,10,30,'chi2='+string(res(0))+ ' DF='+string(df)
    xyouts,10,26,'prob='+string(res(1))
endfor
psdirect,program,ps,/stop
!p.charsize=1
end

```

5. vertaa KS ja U-testejä:

Käytä tehtävässä 2. luettua dataa. Lisätään jokaiseen galaksi-mittaukseen pieni ylimäärä (kutsutaan saatua otosta “GAL+”): kuinka suuri tämän (keinotekoisen) lisäyksen tulee olla, jotta

a) KS-testi ja

b) U-testi

antaisi aiheen hylätä olettamus ettei “GAL+” ja “SKY” mittausten valilla ole eroa? (Huom: onko kyseessä 1- vai 2-suuntainen testi?)


c) Totea vielä millainen vaikutus on sillä jos kerrot jokaisen galaksi-mittauksen arvon jollakin vakiolla (mittausten hajonta kasvaa tällä tekijällä). Vertaa KS- ja U-testejä. (Huom: onko kyseessä 1- vai 2-suuntainen testi?)

Esimerkkiohjelma `idlharj5_kstwo_WJ_cont.pro` soveltaa KS- ja U-testia 1) alkuperäiseen dataan, 2) keinotekoiseen dataan jossa on lisetty 0.5 mJy GAL otokseen, 3) keinotekoiseen dataan jossa GAL otoksen datat kerrottu 1.3:lla


```
IDL> .run idlharj5_kstwo_WJ_cont.pro
WJ EXERCISE 5.1/5.2
-----
kstwo:
Maximum D      0.0674111
Probability in two tailed test      0.426520
chi2-statistic      3.00997
probability in one-tailed test:      0.222021
-----
rs_test:
Z      0.341223
Probability in one-tailed test      0.366468

WJ EXERCISE 5.1/5.2 + artificially added 0.5 to galaxy data
-----
kstwo:
Maximum D      0.0814543
Probability in two tailed test      0.212351
chi2-statistic      4.39468
probability in one-tailed test:      0.111098
-----
rs_test:
Z      1.65319
Probability in one-tailed test      0.0491462

WJ EXERCISE 5.1/5.2 + artificially spread galaxy data by 1.3
-----
kstwo:
Maximum D      0.113239
Probability in two tailed test      0.0262804
chi2-statistic      8.49361
probability in one-tailed test:      0.0143099
-----
rs_test:
Z      0.575005
Probability in one-tailed test      0.282644
```




Alkuperäinen data: ei eroja GAL ja SKY



Lisätty GAL+0.5: (1-puolinen testi)
U-testin $p = 0.049 \Rightarrow$ ero tulee esille
KS-testi $p = 0.111$ ei merkittävä

Jos lisättäisiin GAL+1.0
 \Rightarrow ero tulisi esille molemmissa testeissä
U-testi $p = 0.0013$
KS-testi $p = 0.01$



kerrottu GAL*1.3 (2-puolinen testi)
 \Rightarrow ero tulee esille KS-testissä

```

program0='idlharj5_kstwo_WJ_cont' & ps=-1

for icase=1,3 do begin
;-----
;read WJ data in 5point2.dat
;-----
    file='5point2.dat'
    close,1
    openr,1,file
    line=''
    readf,1,line
    readf,1,line
    gal=fltarr(386)
    readf,1,gal
    readf,1,line
    readf,1,line
    readf,1,line
    sky=fltarr(290)
    readf,1,sky
    close,1
;-----
;make an artificial offset to galaxy-data?
    if(icase eq 1) then begin
        program=program0+'_a'
        title='WJ EXERCISE 5.1/5.2'
    endif
    if(icase eq 2) then begin
        program=program0+'_b'
        title='WJ EXERCISE 5.1/5.2 + artificially added 0.5 to galaxy data'
        gal=gal+0.5
    endif
    if(icase eq 3) then begin
        program=program0+'_c'
        title='WJ EXERCISE 5.1/5.2 + artificially spread galaxy data by 1.3'
        gal=gal*1.3
    endif
    psdirect,program,ps,/col
    coll=1
    if(!d.name eq 'PS') then coll=0
;-----
;plot histograms
;histo_f,x,x1,x2,dx,xx,yy,gg
    histo_f,gal,-20,20,1,xt,yt_gal,/nos
    histo_f,sky,-20,20,1,xt,yt_sky,/nos
    nwin
    plot,xt,yt_gal,xtitle='flux density mJy',ytitle='Number',psym=10,$
        title=title,yr=[0,50],ys=1,thick=2
    oplot,xt+.02,yt_sky+.2,psym=10,lines=2,col=2,thick=3
    label_data,0.55,0.95,['GALAXY','SKY'],lines=[0,2],col=[coll,2]

    print,''
    print,title
;KSTWO
; Return the Kolmogorov-Smirnov statistic and associated 2-tailed probability
; that two arrays of data values are drawn from the same distribution
    kstwo,gal,sky,d,prob
    print,'-----'
    print,'kstwo:'
    print,'Maximum D',d
    print,'Probability in two tailed test',prob
;One-tailed test-statistic
    m=n_elements(gal)
    n=n_elements(sky)
    chi2=4*d^2*m*n/(m+n)
    print,'chi2-statistic', chi2
    print,'probability in one-tailed test:', 1-chisqr_pdf(chi2,2)

```

```

xyouts,-18,38,'KS-test: (kstwo.pro)'
xyouts,-18,36,'max D'+string(d,'(f6.3)')
xyouts,-18,34,'1-tailed significance: (gal>sky)'
xyouts,-16,32,string(1-chisqr_pdf(chi2,2),'(f6.3)')

if(icase eq 3) then begin
    xyouts,-18,28,'2-tailed significance: (gal ne sky)'
    xyouts,-16,26,string(prob,'(f6.3)')
endif
;-----
;Apply Wilcoxon-Mann-Whitney test
; RS_TEST(X,Y) tests the hypothesis that two sample populations,
; {X[i], Y[i]}, have the same mean of distribution against the
; hypothesis that they differ. The result is a two-element vector
; containing the nearly-normal test statistic Z and the one-tailed
; probability of obtaining a value of Z or greater. This type of
; test is often referred to as the Wilcoxon Rank-Sum Test or Mann-
; Whitney U-Test.

res=rs_test(sky,gal)
Z=res(0)
prob=res(1)
print,'-----'
print,'rs_test:'
print,'Z',Z
print,'Probability in one-tailed test',prob

xyouts, 8,38,'U-test: (rs_test.pro)'
xyouts, 8,36,'Z'+string(Z,'(f6.3)')
xyouts, 8,34,'1-tailed significance: (gal>sky)'
xyouts, 10,32,string(prob,'(f6.3)')

psdirect,program,ps,/stop
endfor
end

```

6. χ^2 -testisuuren käyttäminen (Gregory'n kirjan esimerkki 7.2.1)

Tiedostossa “gregory7point1.dat” on erään radiogalaksin vuontiheyden mittauksia eri ajanhetkiltä. Ovatko mittaukset sopusoinnussa sen kanssa, että galaksin vuontiheys on vakio, vai onko syytä olettaa että vuontiheys vaihtelee ajan mukana? Mittauslaitteistosta tiedetään, että mittausvirheet ovat Gaussisesti jakaantuneet (keskiarvo 0, hajonta 2.7 mJy).

a) Lue data

b) Muodosta χ^2 -testisuure ja testaa hypoteesia, jonka mukaan vuo on vakio.
